
1 Introduzione 
Il presente documento ha lo scopo primario di fornire all’utente una panoramica della logica di compilazione dello script ATE 
e secondariamente mostrare alcuni esempi applicativi. Le informazioni, in esso contenute sono destinate ad un installatore 
certificato secondo il programma di formazione Bosch Security Academy. Per maggiori informazioni: 
https://academy.globalteach.com/ST/Login.aspx  
 
Il documento non è sostitutivo al manuale ufficiale, ma un importante supporto alla compressione di quest’ultimo, scaricabile 
dal seguente collegamento: download 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 

https://academy.globalteach.com/ST/Login.aspx
https://resources-boschsecurity-cdn.azureedge.net/public/documents/ATSL_7.60__Special_all_23843870091.pdf

2 Le Basi dell’Alarm Task Editor  

2.1 Variabili Comuni 
VCARule(1,1) = Regola di videoanalisi 1 (VCARule(x,1)) del canale video 1 (VCARule(1,x)) 
VCARule(1,1,1) = simile al precedente ma utilizzabile, il terzo valore (VCARule(x,x,1))  rappresenta il profilo VCA 
Input(1) = Ingresso 1 della telecamera 
Relay(1) = Uscita relè 1 della telecamera 
Tempstate(1) = Tempstate 1 (vedi capitolo riguardante) 
VirtualAlarm(1)= Virtual Alarm 1 (vedi capitolo riguardante) 
IsActivated()= calendario (vedi capitolo riguardante) 

2.2 Negazione Variabili 
! = Not = significa non 
!VCARule(1,1) = posto davanti la condizione, la rende da vera a falsa 

2.3 Congiunzione variabili 
&& = And = significa e 
VCARule(1,1) && VCARule(1,2) = la condizione risultante è vera solo se entrambe sono vere 
 
|| = Or =  significa o 
VCARule(1,1) || VCARule(1,2) = la condizione risultante è vera se una delle 2 condizioni è vera 

2.4 Il ciclo IF 
if(condizione) then Azione1, Azione2 else Azione3, Azione4; 
 
Esempi: 
 
if(VCARule(1,1)) then Relay(1):=true else Relay(1):=false; 
 
Significato = se la VCARule(1,1) è vera attiva il Relè1 altrimenti se la VCARule(1,1) è falsa disattiva il Relè1 
 
if(VCARule(1,1) && !VCARule(1,2)) then Relay(1):=true else Relay(1):=false; 
 
Significato = se VCARule(1,1) è vera e la VCARule(1,2) è falsa (nota il !) allora attiva il relè 1 altrimenti in tutti gli altri 
casi disattiva il relè 1  
 
if(VCARule(1,1)) then Relay(1):=true; 
 
Significato = se la VCARule(1,1) è vera attiva il Relè1 altrimenti in tutti gli altri casi non fare niente 
Per evitare confusione e progettare correttamente la funzione IF, è sempre meglio creare una tabella delle 
condizioni/azioni: 
 
Esempio: 
 
if(VCARule(1,1) && VCARule(1,2)) then Relay(1):=true else Relay(1):=false; 
 
L’AND tra le 2 variabili corrisponde alla tabella: 
 

CONDIZIONE AZIONE   

VCARule(1,1) VCARule(1,2)     

VERO VERO Attiva Relay(1) then 

FALSO VERO Disattiva Relay (1) else 

VERO FALSO Disattiva Relay (1) else 

FALSO FALSO Disattiva Relay (1) else 

2.5 I Tempstate 
È un timer, una condizione vera per un determinato tempo (monostabile) oppure che cambia stato in determinato intervallo 
temporale 
Possono essere definiti sino a 18 Timer all’interno dello script 
 
OperationMode monostable:={High(50)};      
TempState(1):= monostable; 
 
La condizione è vera per 50 decimi di secondo (5 secondi) 
 
OperationMode bistable:={Low(100)High(50)}; 
TempState(1):= bistable; 
 
La condizione è vera per 50 decimi di secondo (5 secondi), falsa per 100 decimi di secondo poi vera per 50…..all’infinito 
 
La variabile che abbiamo definito è TempState(1), può essere utilizzata in un ciclo IF come in esempio: 
 
if(TempState(1)) then Relay(1):=true else Relay(1):=false; 

2.6 I Calendari 
È una variabile vera in un determinato ora/giorno della settimana 
 
Timer daily:={TimeBegin(01:13)TimeEnd(14:30)}; 
 
Calendario giornaliero che inizia alle 01:13 e finisce alle 14:30 
 
Timer weekend := { TimeBegin( 0:00 ) TimeEnd( 23:59 ) DayBegin( Sa ) DayEnd( Su ) }; 
 
Calendario settimanale che inizia alle 00:00 e finisce alle 23:59 da sabato a domenica 
 
I calendari che abbiamo definito sono daily e weekend, posso essere utilizzati in un ciclo if richiamandoli con le variabili 
IsActivated(daily) e IsActivated(weekend) 
 
if(IsActivated(daily) ) then Relay(1):=true else Relay(1):=false; 
 
if(IsActivated(weekend) ) then Relay(1):=true else Relay(1):=false; 

2.7 Scripts GUI 
Sono script che vengono compilati automaticamente seguendo una procedura presente nel menù della telecamera. 
Quando si decide di compilare lo script utilizzando uno o più script GUI, ricordatevi che la procedura sovrascriverà lo script 
ATE presente. 
 
Alcuni esempi di script GUI: 
 

• Pianificatore modalità scena telecamera  
• VCA operation mode su evento attivato  
• Email di allarme  
• Azioni su Ingresso allarme 1 e 2  
• Uscita allarme relè a bordo  

 
NB: terminata la procedura GUI, è comunque possibile entrare nel compilatore dello script ATE e cambiare alcune 
condizioni/azioni dei cicli IF. 
 
Esempio: 
 
Risultato GUI Attivazione relè: 
 
//{{relais_start 
OperationMode modeRel1:={High(10) Idle(open)}; 
Relay(1):=modeRel1; 

if(NightMode)then Relay(1):=true; 
//}}relais_end 65f65d7344c834459f26cd43944b6630 
 
Significato: 
 
OperationMode modeRel1:={High(10) Idle(open)};   
Relay(1):=modeRel1; 
 
Definiamo un comportamento relè che quando è a riposo risulta aperto e se eccitato si chiude per 10 decimi di secondo (1 
secondo). Lo stato iniziale “open” può essere cambianto in “close”, come il tempo di cambio condizione di 10 decimi di 
secondo può essere variato. 
 
if(NightMode)then Relay(1):=true else Relay(1):=false; 
 
la condizione del ciclo if è NightMode (modalità notturna), può essere variata ad esempio mettendo la condizione di 
allarme della VCA:  
 
if(VCARule(1,1))then Relay(1):=true else Relay(1):=false; 

3 Funzionalità Avanzate 

3.1 Invio Immagine allarme a server FTP 
SERVER FTP (HOST: ftp.bosch.it USR: ftpuser PSW: pass) 
 
ATE: 
 
JpegPosting jp1:={IP("ftp.bosch.it") Login("ftpuser")Password("pass") Format(Medium) 
Suffix(Date) Path("/") Camera(1) VCAOverlay(true)}; 
 
if(VCARule(1,1)) then jp1; 
 
ARGOMENTI: 
 
(jp1) nome del comando (può essere cambiato a piacimento) 
(/) è il percorso nel server ftp  
(VCAOverlay(true)) invia l’immagine con sovraimpressi i metadati 
(VCARule(1,1)) è la condizione che attiva 

3.2 RCP Command: Reset task contatori a bordo camera 
ATE: 
 
RcpCommand 
resetCounters:={Command("rcp.xml?command=0x0b4a&type=P_OCTET&direction=WRITE&num=1&payload
=0x00")}; 
 
Timer midnight := {TimeBegin( 00:01) TimeEnd( 00:02 ) };  
if(IsActivated(midnight)) then resetCounters; 
 
if(Input(1))then resetCounters; 
 
ARGOMENTI: 
 
(TimeBegin( 00:01) TimeEnd( 00:02 )) è l’ora esatta di reset automatico dei contatori, in questo caso 00:01, un 
trigger di 1 secondo tra le 00:01 alle 00:02 
(Input(1)) è la condizione che attiva il comando reset manuale 
 
 

ftp://ftp.bosch.it/

3.3 Cambio Modalità di scena da commutazione notte/giorno 
ATE: 
 
BicomCommand cmdModeDay := { Server(CameraSrv) ObjectId(0x0200) Payload("0x0007") };  
BicomCommand cmdModeNight := { Server(CameraSrv) ObjectId(0x0200) Payload("0x0008") };  
if (NightMode) then cmdModeNight else cmdModeDay; 
 
ARGOMENTI: 
 
(07,08) corrispondono al numero della modalità di scena da attivare, nel menù la prima della lista corrisponde a 01, il 
valore è espresso in esadecimale quindi la modalità 10 corrisponde a 0A 
(NightMode) è la condizione che attiva il comando cmdModeNight e corrisponde al passaggio notturno del sensore 
crepuscolare. 

3.4 Aggiornamento automatico dell’immagine di riferimento per la manomissione “Reference Check” 
ATE: 
 
RcpCommand takeRef:= 
{Command("rcp.xml?command=0x0805&type=F_FLAG&direction=WRITE&num=1&payload=0x01")}; 
 
OperationMode trigger3600sec := {Low(36000)High(1) }; 
  
TempState(1):=trigger3600sec; 
if (TempState(1)) then takeRef; 
 
ARGOMENTI: 
 
(36000) è il tempo da impostare per prendere il riferimento espresso in decimi di secondo. In questo caso il riferimento 
verrà preso dopo 3600 secondi 

3.5 RCP Command: Virtual Alarm tra due telecamere 
È possibile trasferire l’informazione tra 2 device Bosch presenti all’interno del medesimo network. 
Ogni telecamera può inviare più allarmi virtuali ma ne può ricevere sino a 4. 
 
Telecamera che invia l’informazione (IP:192.168.0.1 Username: service Password:Invio1234!) 
 
ATE: 
 
RcpCommand setRemoteVirtualinput1 := { 
Command("rcp.xml?command=0x0a8b&type=F_FLAG&direction=WRITE&payload=1&num=1")  
IP("192.168.0.2") Password("Ricevo1234!")  UserName (Service)}; 
 
RcpCommand clearRemoteVirtualinput1 := { 
Command("rcp.xml?command=0x0a8b&type=F_FLAG&direction=WRITE&payload=0&num=1")  
IP("192.168.0.2") Password("Ricevo1234!") UserName (Service)}; 
 
if (VCARule(1,1))  then setRemoteVirtualinput1 else clearRemoteVirtualinput1; 
 
ARGOMENTI: 
 
(1) è il numero della funzione ogni telecamera può avere più funzioni e inviare più allarmi virtuali, basta sostituire il numero 
(num=1) è il numero del Virtual Alarm che verrà attivato sulla telecamera remota, ogni telecamera può accettare sino a 4 
allarmi virtuali 
(VCARule(1,1)) è la condizione che attiva il comando 
 
Telecamera che riceve l’informazione (IP:192.168.0.2 Username: service Password:Ricevo1234!) 
 
L’allarme virtuale 1 ricevuto è esplicitato dalla variabile VirtualAlarm(1) e  può essere utilizzata  in un ciclo IF come in 
esempio: 

ATE: 
 
if (VirtualAlarm(1))then Relay(1):=true else Relay(1):=false; 

3.6 RCP Command: Attivazione/Disattivazione audio 
Esempio attivazione audio segue l’Input telecamera 
 
ATE: 
 
RcpCommand audio_on 
:={Command("rcp.xml?command=0x000c&type=F_FLAG&direction=WRITE&num=1&payload=1")}; 
 
RcpCommand audio_off 
:={Command("rcp.xml?command=0x000c&type=F_FLAG&direction=WRITE&num=1&payload=0")}; 
 
if (Input(1)) then audio_on else audio_off; 
 
ARGOMENTI: 
 
(Input(1)) è la condizione che attiva l’audio 

3.7 Invio comandi AUX su AUTODOME 
Esempio autotracking = AUX 78 segue l’Input telecamera 
 
ATE: 
 
//Comando Autotrack ON 
ControlCode auxon78 := { Operation(AuxOn) Number(78) }; 
 
//Comando Autotrack OFF 
ControlCode auxoff78 := { Operation(AuxOff) Number(78) }; 
 
if(Input(1)) then auxoff78 else auxon78; 
 
ARGOMENTI: 
 
(Input(1)) è la condizione che attiva l’AUX 78 

3.8 Attivazione inactivity su AUTODOME 
Esempio inactivity segue l’Input telecamera 
 
ATE: 
 
RcpCommand 
inactivityon:={Command("rcp.xml?command=0x09A5&type=P_OCTET&direction=WRITE&num=0&payload=
0x8000060140030001")}; 
RcpCommand 
inactivityoff:={Command("rcp.xml?command=0x09A5&type=P_OCTET&direction=WRITE&num=0&payload
=0x8000060140030000")}; 
if(Input(1)) then inactivityoff; 
 
ARGOMENTI: 

(Input(1)) è la condizione che attiva l’inactivity 
 

4 HTTP Command 
 
PREMESSA: Per permettere alla telecamera di ricevere http command, è necessario abilitare l’autenticazione basic 
 
I comandi http per le device Bosch sono tradotte nello script ATE in questo modo: 
 
BROWSER: 
 
http://utente:password@IP/argomento 
 
ATE: 
 
HttpCommand nome:= {Command("argomento") Password("password") UserName("utente") 
IP("IP")}; 
 
NB UserName: nello script ATE il nome utente va sempre messo tra "", l’unica eccezione è se il comando viene inviato tra 
telecamere Bosch, dove l’utente è service e quindi può essere inserito senza virgolette e con la prima lettera maiuscola 
esempio: UserName(Service) 

4.1 Invio allarme a modulo Arteco Everywhere I/O  
ATE: 
 
HttpCommand set_LED_Alarm := {Command("set.cgi?set=1&time1=10 ") Password("arteco") 
UserName("admin") IP("192.168.0.163") Port(HTTP) SSL(false)};  
 
if(VCARule(1,1)) then set_LED_Alarm;     
 
ARGOMENTI: 
 
(set_LED_Alarm) nome del comando (può essere cambiato a piacimento es. alarm_ART) 
(set=1) numero dell’output su Arteco (set=1 corrisponde ad output 2 mettere set=2 per output 3 ecc…) 
(time1=10) tempo di attivazione output (time1=10 corrisponde a 1 secondo mettere time1=100 per 10 secondi) 
(arteco) è la password del modulo Arteco che ho impostato 
(admin) è l’utente del modulo Arteco che impostato 
(192.168.0.163) è l’IP del modulo Arteco che ho impostato 
(VCARule(1,1)) è la condizione che attiva il comando (in questo caso il task 1 della videoanalisi) 

4.2 Invio allarme a modulo Advantec  
ATE: 
 
HttpCommand set_LED_Alarm:= { Command("digitaloutput/all/value") 
IP("192.168.0.163")Port(HTTP)SSL(false)Password("00000000")ForceBasicAuth(true)UserName("r
oot")Method(POST)ContentType("text/plain")Payload("DO0=1")}; 
 
if(VCARule(1,1)) then set_LED_Alarm;     
 
ARGOMENTI: 
 
(set_LED_Alarm) nome del comando (può essere cambiato a piacimento es. alarm_ART) 
(DO0=1) numero dell’output e attivazione DO0 è l’output 1 e =1 è l’attivazione, mettere  =0 per disattivare 
(00000000) è la password del modulo  
(root) è l’utente del modulo  
(192.168.0.163) è l’IP del modulo  
(VCARule(1,1)) è la condizione che attiva il comando (in questo caso il task 1 della videoanalisi) 

http://utente:password@IP/argomento

4.3 Invio allarme a tromba IP Bosch 
ATE: 
 
HttpCommand sendHttp_on 
:={Command("api/ext/v1/vgpis/1")SSL(false)Port(80)IP("192.168.10.163")Password("12345678")
UserName("User")Method(POST)ContentType("application/json")Payload("true")ForceBasicAuth(t
rue)}; 
 
if(VCARule(1,1)) then sendHttp_on; 
 
ARGOMENTI: 
 
(sendHttp_on) nome del comando (può essere cambiato a piacimento es. alarm_ART) 
(vgpis/1) numero del messaggio audio registrano vgpis/1 corrisponde al n°1, mettere vgpis/2 per il messaggio 2 
(12345678) è la password della tromba 
(User) è l’utente della tromba  
(192.168.0.163) è l’IP della tromba 
(VCARule(1,1)) è la condizione che attiva il comando (in questo caso il task 1 della videoanalisi) 

4.4 Invio allarme a centrale KSENIA con auto-ripristino gestito dalla centrale  
ATE: 
 
HttpCommand set_ZONE5 := {Command("ZONES/5/ALARM") Port (69) IP("192.168.2.160")}; 
 
if(VCARule(1,1)) then set_ZONE5; 
 
ARGOMENTI: 
 
(set_ZONE5) nome del comando (può essere cambiato a piacimento) 
(5) è la zona in centrale che vogliamo attivare (in questo caso la zona 5) 
(69) è la porta di comunicazione della centrale 
(192.168.2.160) è l’IP della centrale 
(VCARule(1,1)) è la condizione che attiva il comando (in questo caso il task 1 della videoanalisi) 

4.5 Invio allarme a centrale KSENIA con ripristino gestito dalla telecamera 
ATE: 
 
HttpCommand attiva_ZONE5:= {Command("ZONES/5/ALARM")IP("192.168.2.160") Port(69)}; 
 
HttpCommand disattiva_ZONE5:= {Command("ZONES/5/RESTORE")IP("192.168.2.160") Port(69)}; 
 
if (VCARule(1,1)) then attiva_ ZONE5 else disattiva_ ZONE5 ; 
  
ARGOMENTI: 
 
(attiva_ZONE5, disattiva_ZONE5) sono i nomi dei comandi  
(5) è la zona in centrale che vogliamo attivare (in questo caso la zona 5) 
(69) è la porta di comunicazione della centrale 
(192.168.2.160) è l’IP della centrale 
(VCARule(1,1)) è la condizione che attiva il comando (in questo caso il task 1 della videoanalisi) 

4.6 Invio allarme a centrale SATEL con auto-ripristino gestito dalla centrale  
ATE: 
 
HttpCommand vca1:= {Command("zona230")IP("192.168.2.160") Port(7070) SSL(false) }; 
if (VCARule(1,1)) then vca1; 

ARGOMENTI: 
 
(vca1) nome del comando (può essere cambiato a piacimento) 
(zona230) è la zona in centrale che vogliamo attivare (in questo caso la zona 230) 
(192.168.2.160) è l’IP della centrale 
(7070) è la porta di comunicazione della centrale 
(VCARule(1,1)) è la condizione che attiva il comando (in questo caso il task 1 della videoanalisi) 

4.7 Invio comando di attivazione/disattivazione IR a bordo camera 
TELECAMERA RICEVENTE IL COMANDO (IP: 192.168.0.1 USR: service PSW: Service1234!) 
 
BROWSER: 
 
http://service:Service1234!@192.168.0.1/rcp.xml?command=0x09A5&type=P_OCTET&direction=WRITE&num=0&payload
=0x8000040410030001 
 
ATE: 
 
HttpCommand set_IR_ON:= 
{Command("rcp.xml?command=0x09A5&type=P_OCTET&direction=WRITE&num=0&payload=0x800004041003
0001") Password("Service1234!") UserName(Service) IP("192.168.0.1")}; 
 
ARGOMENTI: 
 
(set_IR_ON) nome del comando (può essere cambiato a piacimento) 
(1) corrisponde all’attivazione della funzione IR mettere 0 per disattivare oppure 2 per il funzionamento automatico 

4.8 Invio comando di attivazione/disattivazione modalità notturna 
TELECAMERA RICEVENTE IL COMANDO (IP: 192.168.0.1 USR: service PSW: Service1234!) 
 
BROWSER: 
 
http://service:Service1234!@192.168.0.1/rcp.xml?command=0x09A5&type=P_OCTET&direction=WRITE&num=1&payload
=0x8000040140020001  
 
ATE: 
 
HttpCommand set_NIGHT_ON:= 
{Command("rcp.xml?command=0x09A5&type=P_OCTET&direction=WRITE&num=1&payload=0x800004014002
0001") Password("Service1234!") UserName(Service) IP("192.168.0.1")}; 
 
ARGOMENTI: 
 
(set_NIGHT_ON) nome del comando (può essere cambiato a piacimento) 
(1) corrisponde all’attivazione della funzione mettere 0 per disattivare oppure 2 per il funzionamento automatico 

4.9 Invio comando di attivazione/disattivazione relè a bordo camera 
TELECAMERA RICEVENTE IL COMANDO (IP: 192.168.0.1 USR: service PSW: Service1234!) 
 
BROWSER: 
 
http://service:Service1234!@192.168.0.1/rcp.xml?command=0x01c1&type=F_FLAG&direction=WRITE&payload=1&num=
1 
 
ATE: 
 

http://service:Service1234!@192.168.0.1/rcp.xml?command=0x0ae4&type=T_OCTET&direction=WRITE&payload=2&num=1
http://service:Service1234!@192.168.0.1/rcp.xml?command=0x0ae4&type=T_OCTET&direction=WRITE&payload=2&num=1
http://service:Service1234!@192.168.0.1/rcp.xml?command=0x0ae4&type=T_OCTET&direction=WRITE&payload=2&num=1
http://service:Service1234!@192.168.0.1/rcp.xml?command=0x0ae4&type=T_OCTET&direction=WRITE&payload=2&num=1
http://service:Service1234!@192.168.0.1/rcp.xml?command=0x01c1&type=F_FLAG&direction=WRITE&payload=1&num=1
http://service:Service1234!@192.168.0.1/rcp.xml?command=0x01c1&type=F_FLAG&direction=WRITE&payload=1&num=1

HttpCommand set_RL_ON:= 
{Command("rcp.xml?command=0x01c1&type=F_FLAG&direction=WRITE&payload=1&num=1") 
Password("Service1234!") UserName(Service) IP("192.168.0.1")}; 
 
ARGOMENTI: 
 
(set_RL_ON) nome del comando (può essere cambiato a piacimento) 
(payload=1) corrisponde all’attivazione della funzione relè mettere payload=0 per disattivare la funzione 

4.10 Invio comando di attivazione/disattivazione allarme virtuale 
TELECAMERA RICEVENTE IL COMANDO (IP: 192.168.0.1 USR: service PSW: Service1234!) 
 
BROWSER: 
 
http://service:Service1234!@192.168.0.1/rcp.xml?command=0x0a8b&type=F_FLAG&num=1&direction=WRITE&payload
=1 
 
ATE: 
 
HttpCommand set_IN_ON:= 
{Command("rcp.xml?command=0x0a8b&type=F_FLAG&num=1&direction=WRITE&payload=1") 
Password("Service1234!") UserName(Service) IP("192.168.0.1")}; 
 
ARGOMENTI: 
 
(set_IN_ON) nome del comando (può essere cambiato a piacimento) 
(num=1) è il numero del Virtual Alarm che verrà attivato sulla telecamera remota, ogni telecamera può accettare sino a 4 
allarmi virtuali 
(payload=1) corrisponde all’attivazione della funzione Virtual Alarm mettere payload=0 per disattivare la funzione 

4.11 Invio comando di preset 1 su Autodome 
TELECAMERA RICEVENTE IL COMANDO (IP: 192.168.0.1 USR: service PSW: Service1234!) 
 
BROWSER: 
 
http://service:Service1234!@192.168.0.1/rcp.xml?command=0x09A5&type=P_OCTET&direction=WRITE&num=1&payload
=0x80000201B080070501 
 
ATE: 
 
HttpCommand set_PRE1:= 
{Command("rcp.xml?command=0x09A5&type=P_OCTET&direction=WRITE&num=1&payload=0x80000201B080
070501") Password("Service1234!") UserName(Service) IP("192.168.0.1")}; 
 
ARGOMENTI 
 
(set_PRE1) nome del comando (può essere cambiato a piacimento) 
(01) è il preset 1 espresso in esadecimale, per il preset 10, ad esempio, il corrispondente esadecimale è 0A 

4.12 Invio comando di attivazione/disattivazione del tracking (Autodome serie 7000) 
TELECAMERA RICEVENTE IL COMANDO (IP: 192.168.0.1 USR: service PSW: Service1234!) 
 
BROWSER: 
 
http://service:Service1234!@192.168.0.1/rcp.xml?command=0x09A5&type=P_OCTET&direction=WRITE&num=1&payload
=0x80000201B08007014E 
 
ATE: 

http://service:Service1234!@192.168.0.1/rcp.xml?command=0x0a8b&type=F_FLAG&num=1&direction=WRITE&payload=1
http://service:Service1234!@192.168.0.1/rcp.xml?command=0x0a8b&type=F_FLAG&num=1&direction=WRITE&payload=1
http://service:Service1234!@192.168.0.1/rcp.xml?command=0x09A5&type=P_OCTET&direction=WRITE&num=1&payload=0x80000201B080070501
http://service:Service1234!@192.168.0.1/rcp.xml?command=0x09A5&type=P_OCTET&direction=WRITE&num=1&payload=0x80000201B080070501
http://service:Service1234!@192.168.0.1/rcp.xml?command=0x09A5&type=P_OCTET&direction=WRITE&num=1&payload=0x80000201B08007014E
http://service:Service1234!@192.168.0.1/rcp.xml?command=0x09A5&type=P_OCTET&direction=WRITE&num=1&payload=0x80000201B08007014E

HttpCommand trackON:= 
{Command("rcp.xml?command=0x09A5&type=P_OCTET&direction=WRITE&num=1&payload=0x80000201B080
07014E") Password("Service1234!") UserName(Service) IP("192.168.0.1")}; 
 
ARGOMENTI 
 
(1) corrisponde all’attivazione, mettere 2 per disattivare 
(trackON) nome del comando (può essere cambiato a piacimento) 

4.13 Invio comando di attivazione/disattivazione del tour (Autodome) 
TELECAMERA RICEVENTE IL COMANDO (IP: 192.168.0.1 USR: service PSW: Service1234!) 
 
BROWSER: 
 
http://service:Service1234!@192.168.0.1/rcp.xml?command=0x09A5&type=P_OCTET&direction=WRITE&num=1&payload
=0x80000201B080070108 
 
ATE: 
 
HttpCommand tourON:= 
{Command("rcp.xml?command=0x09A5&type=P_OCTET&direction=WRITE&num=1&payload=0x80000201B080
070108") IP("192.168.0.1")}; 
 
ARGOMENTI: 
 
(1) corrisponde all’attivazione, mettere 2 per disattivare 
(tourON) nome del comando (può essere cambiato a piacimento) 

4.14 Invio comandi di cambio profilo video analisi 
TELECAMERA RICEVENTE IL COMANDO (IP: 192.168.0.1 USR: service PSW: Service1234!) 
 
BROWSER: 
 
http://service:Service1234!@192.168.0.1/rcp.xml?command=0x0A65&type=T_OCTET&direction=WRITE&num=1&payloa
d=0 
 
ATE: 
 
HttpCommand profilo:= 
{Command("rcp.xml?command=0x0A65&type=T_OCTET&direction=WRITE&num=1&payload=0") 
Password("Service1234!") UserName(Service) IP("192.168.0.1")}; 
 
ARGOMENTI: 
 
(0) corrisponde al profilo silent, mettere 1 per il profilo 1, 2 per il profilo 2 e 0xfd per il profilo OFF  
(profilo) nome del comando (può essere cambiato a piacimento) 

4.15 Maschera globale di privacy (solo su telecamere fisse con FW>6.0) 
TELECAMERA RICEVENTE IL COMANDO (IP: 192.168.0.1 USR: service PSW: Service1234!) 
 
BROWSER: 
 
http://service:Service1234!@192.168.0.1/rcp.xml?command=0x0BD8&type=P_OCTET&direction=WRITE&num=1&payloa
d=0x0008000000153fe9001c0001010004000000000000000000800000008000800000008000 
 

http://service:Service1234!@192.168.0.1/rcp.xml?command=0x09A5&type=P_OCTET&direction=WRITE&num=1&payload=0x80000201B080070108
http://service:Service1234!@192.168.0.1/rcp.xml?command=0x09A5&type=P_OCTET&direction=WRITE&num=1&payload=0x80000201B080070108
http://service:Service1234!@192.168.0.1/rcp.xml?command=0x0A65&type=T_OCTET&direction=WRITE&num=1&payload=0
http://service:Service1234!@192.168.0.1/rcp.xml?command=0x0A65&type=T_OCTET&direction=WRITE&num=1&payload=0
http://service:Service1234!@192.168.0.1/rcp.xml?command=0x0A65&type=T_OCTET&direction=WRITE&num=1&payload=0
http://service:Service1234!@192.168.0.1/rcp.xml?command=0x0A65&type=T_OCTET&direction=WRITE&num=1&payload=0

ATE: 
 
HttpCommand mask:= 
{Command("rcp.xml?command=0x0BD8&type=P_OCTET&direction=WRITE&num=1&payload=0x000800000015
3fe9001c0001010004000000000000000000800000008000800000008000") Password("Service1234!") 
UserName(Service) IP("192.168.0.1")}; 

4.16 Attivazione custom tour su PTZ 
TELECAMERA RICEVENTE IL COMANDO (IP: 192.168.0.1 USR: service PSW: Service1234!) 
 
BROWSER: 
 
http://service:Service1234!@192.168.0.1/rcp.xml?command=0x09A5&type=P_OCTET&direction=WRITE&num=1&payloa
d=0x80000201B080070107 
 
ATE: 
 
HttpCommand ctour:= 
{Command("rcp.xml?command=0x09A5&type=P_OCTET&direction=WRITE&num=1&payload=0x80000201B080
070107") Password("Service1234!") UserName(Service) IP("192.168.0.1")}; 
 
ARGOMENTI: 
 
(1) corrisponde all’attivazione, mettere 2 per disattivare 
(ctour) nome del comando (può essere cambiato a piacimento) 
 

4.17 Attivazione privacy motion blur (se supportato) 
TELECAMERA RICEVENTE IL COMANDO (IP: 192.168.0.1 USR: service PSW: Service1234!) 
 
BROWSER: 
 
http://service:Service1234!@192.168.0.1/rcp.xml?command=0x0bc8&type=T_DWORD&direction=WRITE&payload=3&nu
m=1 
 
ATE: 
 
HttpCommand blur:= 
{Command("rcp.xml?command=0x0bc8&type=T_DWORD&direction=WRITE&payload=3&num=1") 
Password("Service1234!") UserName(Service) IP("192.168.0.1")}; 
 
ARGOMENTI: 
 
(3) corrisponde all’attivazione, mettere 0 per disattivare 
(blur) nome del comando (può essere cambiato a piacimento) 
(num=1) è lo stream 1 

4.18 Richiesta di rimessa a fuoco  
TELECAMERA RICEVENTE IL COMANDO (IP: 192.168.0.1 USR: service PSW: Service1234!) 
 
BROWSER: 
 
http://service:Service0!TR@192.168.0.106/rcp.xml?command=0x09A5&type=P_OCTET&direction=WRITE&num=0&paylo
ad=0x80000401C9020001 

http://service:Service1234!@192.168.0.1/rcp.xml?command=0x09A5&type=P_OCTET&direction=WRITE&num=1&payload=0x80000201B080070107
http://service:Service1234!@192.168.0.1/rcp.xml?command=0x09A5&type=P_OCTET&direction=WRITE&num=1&payload=0x80000201B080070107
http://service:Service1234!@192.168.0.1/rcp.xml?command=0x0bc8&type=T_DWORD&direction=WRITE&payload=3&num=1
http://service:Service1234!@192.168.0.1/rcp.xml?command=0x0bc8&type=T_DWORD&direction=WRITE&payload=3&num=1
http://service:Service0!TR@192.168.0.106/rcp.xml?command=0x09A5&type=P_OCTET&direction=WRITE&num=0&payload=0x80000401C9020001
http://service:Service0!TR@192.168.0.106/rcp.xml?command=0x09A5&type=P_OCTET&direction=WRITE&num=0&payload=0x80000401C9020001

ATE: 
 
HttpCommand refocus:= 
{Command("rcp.xml?command=0x09A5&type=P_OCTET&direction=WRITE&num=0&payload=0x80000401C902
0001") Password("Service1234!") UserName(Service) IP("192.168.0.1")}; 
 
ARGOMENTI: 
 
(refocus) nome del comando (può essere cambiato a piacimento) 

5 HTTPS Command 
 
PREMESSA: Utilizzando i comandi in HTTPS, l’autenticazione avverrà in digest 
 
I comandi http per le telecamere Bosch sono tradotti nello script ATE nel seguente modo: 
 
BROWSER: 
 
https://utente:password@IP/argomento 
 
ATE: 
 
HttpCommand nome:= {Command("argomento") SSL(true)Port(HTTPS) Password("password") 
UserName(utente) IP("IP")}; 
 
I comandi HTTP del capitolo precedente possono essere convertiti in HTTPS semplicemente aggiungendo 
SSL(true)Port(HTTPS) 
 
Tutte le telecamere con firmware superiore al 7.5 accettano comandi HTTPS. 

6 Esempi Applicativi 

6.1 Attivazione relè in caso di allarme di movimento o perdita ethernet 
ATE: 
 
if( ( Motion(1) || !EthernetLink(0) ) ) then Relay(1) := true; 
 
In questo esempio, viene attivato il relè a bordo camera quando si verifica un allarme di movimento o il collegamento 
Ethernet viene interrotto. 

6.2 Invio JPG a server FTP all’attivazione dell’input 1    
ATE: 
 
// Settings for JPEG posting 
JpegPosting errorPosting := { IP( "192.168.0.1" ) Login( "anonymous" ) Password( 
"anonymous" ) Format( Large ) FileName( "alarm_input_1" ) Suffix( Increment ) Camera( All 
) }; 
 
// Define Operation Mode to periodic, 9.9 seconds low and 0.1 second high 
OperationMode periodicJpegPosting := { Low(99) High(1) }; 
 
// On alarm input 1 assign Operation Mode to TempState 
if( Input(1) ) then TempState(1) := periodicJpegPosting else TempState(1) := Stop( 
periodicJpegPosting ); 
 
// Post JPEG when TempState is high (i.e. every 10 seconds) 
if( TempState(1) ) then errorPosting; 
 
In questo esempio, l’invio JPEG viene avviato per tutte le telecamere se l'ingresso allarme 1 è abilitato. 

https://utente:password@IP/argomento

I JPEG vengono pubblicati ogni 10 secondi fino a quando l'input non viene nuovamente cancellato. 

6.3 Attivazione relè di una telecamera remota quando si genera un allarme di movimento  
ATE: 
 
// Define Rcp commands to set and reset Relay 1 at device with IP 192.168.0.1 
RcpCommand setRemoteRelay1 := { 
Command("rcp.xml?command=0x01c1&type=F_FLAG&direction=WRITE&payload=1&num=1") 
IP("192.168.0.1") }; 
RcpCommand clearRemoteRelay1 := { 
Command("rcp.xml?command=0x01c1&type=F_FLAG&direction=WRITE&payload=0&num=1") 
IP("192.168.0.1") }; 
 
// Set TempState to monostable mode, 1 second 
OperationMode monostable := { High(10) }; 
TempState(1) := monostable; 
 
// Set TempState on motion alarm camera 1 
if( Motion(1) ) then TempState(1) := true; 
 
// Set/Reset Relay 
if( TempState(1) ) then setRemoteRelay1 else clearRemoteRelay1; 
 
In questo esempio viene permutato lo stato di un relè su un dispositivo remoto quando viene rilevato un movimento sul 
dispositivo locale. 
Lo stato del relè viene cambiato per 1 secondo e quindi azzerato di nuovo 

6.4 Avvio registrazione di una telecamera remota quando si genera una manomissione 
ATE: 
    
// Define Rcp commands to start and stop recording at device with IP 192.168.0.1 
RcpCommand startRemoteRecording := { 
Command("rcp.xml?command=0x0913&type=F_FLAG&direction=WRITE&payload=1&num=1") 
IP("192.168.0.1") }; 
RcpCommand stopRemoteRecording := { 
Command("rcp.xml?command=0x0914&type=F_FLAG&direction=WRITE&payload=1&num=1") 
IP("192.168.0.1") }; 
 
// Start/Stop recording on certain kinds of tampering camera 1 
if( SignalTooBright(1) || SignalTooDark(1) || SignalLoss(1) ) then startRemoteRecording 
else stopRemoteRecording; 
 
In questo esempio, la registrazione su un dispositivo remoto viene avviata se vengono rilevati determinati tipi di 
manomissione sul dispositivo locale. 
Quando non viene più rilevata alcuna manomissione, la registrazione sul dispositivo remoto viene interrotta. 
Nota: per abilitare il rilevamento delle manomissioni, è necessario attivare l'algoritmo Motion + o IVMD. 

6.5 Attivazione relè se l’input 1 è attivo da troppo tempo 
ATE: 
 
// Set TempState to monostable mode, 5 seconds 
OperationMode monostable := { High(50) }; 
TempState(1) := monostable; 
 
// Initialize Relay(1) on startup 
if( true ) then Relay(1):=false; 
 
// Set TempStates on Input 1 
if( Input(1) ) then TempState(1):=true, TempState(2):=true else TempState(2):=false; 

// Set relay when TempState(1) is cleared after 5 seconds and TempState(2), i.e. Input(1), 
is still enabled. 
if( TempState(2) && !TempState(1) ) then Relay(1):=true; 
 
// Clear Relay with Input(2) 
if( Input(2) ) then Relay(1):=false; 
 
In questo esempio viene permutato lo stato di un relè se un ingresso esterno è abilitato e non azzerato entro un tempo 
specificato. 
Questo potrebbe ad es. essere utilizzato per attivare un allarme se una porta non viene chiusa correttamente entro un 
determinato tempo. 
 
Nota: TempState (2) viene utilizzato poiché gli stati non vengono modificati fino a quando non viene verificato lo script 
completo. 
Quindi tutti gli stati vengono modificati contemporaneamente e lo script viene nuovamente elaborato con i nuovi stati. 
Di conseguenza, l'istruzione 'if (Input (1) &&! TempState (1))' sarebbe vera non appena Input (1) è 
abilitato perché TempState (1) non è impostato finché lo script completo non è stato elaborato. 

6.6 Attiva la registrazione se un allarme di movimento non riceve l’input 1 dopo 5 secondi 
ATE: 
 
// Define recording, camera 1 
Recording alarmRecording := { Camera(1) }; 
 
// Set TempState to monostable mode, 5 seconds 
OperationMode monostable := { High(50) }; 
TempState(1) := monostable; 
 
// Set TempStates on motion alarm from camera 2 
if( Motion(2) ) then TempState(1):=true, TempState(2):=true; 
 
// Clear TempState(2) on Input(1) 
if( Input(1) ) then TempState(2):=false; 
 
// Start recording when alarm notification (TempState(2)) is not cleared within 5 seconds. 
if( TempState(2) && !TempState(1) ) then alarmRecording; 
 
// Stop recording on Input(2) 
if( Input(2) ) then Stop(alarmRecording), TempState(2):=false; 
 
In questo esempio la registrazione viene avviata se si verifica un allarme di movimento e non si risponde agendo 
sull'ingresso 1 entro 5 secondi. 

6.7 Attivare un tempo di anti rimbalzo tra gli allarmi 
ATE: 
 
// Set Relay to monostable mode, 1 second 
OperationMode modeRel1 := { High(10) }; 
Relay(1) := modeRel1; 
 
// Define debounce time, 5 seconds 
OperationMode debounceTime := { High(50) }; 
TempState(2) := debounceTime; 
 
// Set TempState on motion alarm camera 1  
if( Motion(1) ) then TempState(1) := true; 
 
// Trigger Relay and 'debounce timer' on motion alarm if timer is false 
if( TempState(1) && !TempState(2) ) then Relay(1):=true, TempState(2):=true; 

// Reset 'motion event' 
if( TempState(1) ) then TempState(1):=false; 
 
Con l'aiuto di questo esempio, un allarme di movimento può essere "rimbalzato". Cioè, se si verifica un allarme di 
movimento, nessun ulteriore allarme viene generato per questa telecamera entro un tempo specificato. 
Se si verifica un allarme di movimento, viene permutato lo stato di un relè per 1 secondo. Se entro 5 secondi viene 
generato un ulteriore allarme per la stessa telecamera il relè non si attiva nuovamente. 
Nota: TempState (1) viene utilizzato per creare un evento breve quando lo stato di Motion (1) cambia da falso a 
vero. In caso contrario, il relè verrebbe ripristinato se c'è ancora movimento dopo che è trascorso il tempo di antirimbalzo. 

6.8 Inviare un allarme email se il Software Seal viene rotto 
ATE: 
 
// Define RCP message to check for (CONF_CONFIG_SEALING_STATUS): 
MessageCheck check_config_seal := { Message("rcp.xml?message=0x0c8c") Payload("0x02") }; 
 
// Define action to perform if seal is broken (send an alarm e-mail): 
AlarmMail send_mail := { IP("192.168.0.1") To("to@test.com") From("from@test.com") 
Layout(Standard) Login("user") Password("password") Subject("Configuration changed!") }; 
 
// Define a timer for message polling: 
OperationMode poll_5sec := { Low(1) High(49) }; // every 5 seconds (100ms low, 4900ms 
high) 
TempState(1) := poll_5sec; 
 
// Poll for messages: 
if( TempState(1) ) then check_config_seal; 
 
// Check if the event has been received and perform according action 
if( IsActivated(check_config_seal) ) then send_mail; 
 
In questo esempio viene inviata un'e-mail di allarme nel caso in cui il sigillo del software sia rotto. 
Notare che la sigillatura del software deve essere abilitata prima. 
 
Con l'aiuto di un meccanismo di polling viene verificata la ricezione del messaggio RCP 
CONF_CONFIG_SEALING_STATUS. 
Se il messaggio viene ricevuto e il suo payload inizia con "0x02", verrà eseguita l'azione corrispondente. Cioè verrà inviata 
una e-mail di allarme. 

6.9 Attivazione tracking Autodome su allarme VCA 
ATE: 
 
//In caso di allarme VCA è possibile inserire il Traking per seguire l'oggetto in 
movimento. 
//In caso di inattività è necessario impostare il tempo desiderato, nel menu impostazioni 
telecamera 
//Quando la telecamera torna al preset 1 o riparte in tour dal preset 1, viene 
disabilitato il tracking per evitare che intervenga prima della video analisi 
 
//{{alarmrules_start 
//Comando Autotrack ON 
ControlCode auxon78 := { Operation(AuxOn) Number(78) };   
 
//Comando Autotrack OFF 
ControlCode auxoff78 := { Operation(AuxOff) Number(78) };  
 
//Evento di video analisi che attiva Autotrack 
if (Motion(1)) then auxon78;      
 
//Verifica arrivo preset 1 da inattività o su tour 

MessageCheck pp1:={Message("rcp.xml?message=0x09a5") Num(1) 
Payload("0x000601307000010002")};   
 
//Verifica dello stato della posizione ogni 500 ms 
OperationMode msgchk:={Low(5)High(5)};      
TempState(1):=msgchk; 
if(TempState(1)) then pp1;    
 
//Se su preset, Autorack OFF 
if(IsActivated(pp1)) then auxoff78;     
 
//}}alarmrules_end af31e7aaaa8e38049030c40c52b89b6a 
 
In questo esempio, è presente una Autodome che esegue un tour e effettua l’analisi video su preset; 
Se scatta un allarme VCA, viene attivato l’autotrack e al timeout dell’autotrack la telecamera torna al tour. 
Arrivata al preset 1 lo script disabilita l’autotrack. 

7 Esempi di Applicazioni con più telecamere 

7.1 Telecamere, modulo Arteco Everywhere I/O, sensore esterno, server FTP   
DESCRIZIONE AMBIENTE 
 
Si vuole realizzare un impianto, con 2 telecamere configurate con una regola di video analisi e un sensore collegato alla 
seconda telecamera. All’attivarsi di un evento di allarme sia della video analisi che del sensore, le telecamere dovranno 
inviare un’immagine su un server FTP e attivare l’output dedicato del modulo Arteco I/O Everywhere. 
 
IMPOSTAZIONI/FUNZIONALITA’ APPARATI: 
 
TELECAMERA 1 (IP 192.168.0.1 - USR: service - PSW: ServiceTLC01!) 
 

- Quando si genera un allarme di video analisi VCARule(1,1) attiva l’output 1 del modulo Arteco e invia un’immagine 
con il metadato al server FTP 

- All’attivazione dell’allarme virtuale, invia un’immagine con il metadato al server FTP 
 
TELECAMERA 2 (IP 192.168.0.2 - USR: service - PSW: ServiceTLC02!) 
 

- Riceve all’ingresso 1 (Input(1)) l’allarme del sensore esterno 
- Quando si genera un allarme di video analisi VCARule(1,1) attiva l’output 2 del modulo Arteco e invia un’immagine 

con il metadato dell’evento al server FTP 
- Quando attiva l’allarme del sensore all’ Input(1) attiva l’output 3 del modulo Arteco invia un’immagine con il 

metadato al server FTP e attiva un allarme virtuale sulla TLC01 
 

ARTECO (IP 192.168.0.10 - USR: service - PSW: arteco) 
 

- riceve gli allarmi dalle telecamere 
 
SERVER FTP (HOST: ftp.bosch.it USR: ftpuser PSW: pass) 
 

- riceve le immagini inviate dalle telecamere 
 
ATE TLC 01  
 
//Attivazione output 1 Arteco per 1 secondo 
HttpCommand set_LED_Alarm1 := {Command("set.cgi?set=0&time1=10 ") Password("arteco") 
UserName(Service) IP("192.168.0.10")};  
 
//Indirizzo e credenziali server FTP 

ftp://ftp.bosch.it/

JpegPosting jp1:={IP("ftp.bosch.it") Login("ftpuser")Password("pass") Format(Medium) 
Suffix(Date) Path("/") Camera(1) VCAOverlay(true)}; 
 
//Azioni da allarme VCA 
if(VCARule(1,1)) then set_LED_Alarm1, jp1;     
 
//Azioni da attivazione VirtualAlarm(1) 
if(VirtualAlarm(1)) then jp1;   
 
ATE TLC 02  
 
//Attivazione output 2 Arteco per 1 secondo 
HttpCommand set_LED_Alarm2 := {Command("set.cgi?set=1&time1=10 ") Password("arteco") 
UserName(Service) IP("192.168.0.10")};  
 
//Attivazione output 3 Arteco per 1 secondo 
HttpCommand set_LED_Alarm3 := {Command("set.cgi?set=2&time1=10 ") Password("arteco") 
UserName(Service) IP("192.168.0.10")};  
 
//VirtualAlarm 1 su TLC01 
 
RcpCommand setRemoteVirtualinput1 := { 
Command("rcp.xml?command=0x0a8b&type=F_FLAG&direction=WRITE&payload=1&num=1")  
IP("192.168.0.1") Password("ServiceTLC01!")  UserName (Service)}; 
RcpCommand clearRemoteVirtualinput1 := { 
Command("rcp.xml?command=0x0a8b&type=F_FLAG&direction=WRITE&payload=0&num=1")  
IP("192.168.0.1") Password("ServiceTLC01!") UserName (Service)}; 
 
//Indirizzo e credenziali server FTP 
JpegPosting jp1:={IP("ftp.bosch.it") Login("ftpuser")Password("pass") Format(Medium) 
Suffix(Date) Path("/") Camera(1) VCAOverlay(true)}; 
 
//Azione VirtualAlarm su TLC01 
if (Input(1))  then setRemoteVirtualinput1 else clearRemoteVirtualinput1; 
 
//Azioni da allarme VCA 
if(VCARule(1,1)) then set_LED_Alarm2, jp1;     
 
//Azioni da attivazione Input(1) 
if(Input(1)) then set_LED_Alarm3, jp1; 

8 Tools 
Bosch Security Systems rende disponibile un catalogo elettronico volto ad offrire numerose informazioni di prodotto quali: 

• Tool online per supporto alla progettazione 
• Guide agli accessori 
• Webinar 
• Note tecniche 
• Approfondimenti 

 
Download Bosch Catalog – Catalogo elettronico Bosch Security Systems 
 
L’ultima versione di questo manuale, assieme ad altri utili approfondimenti, sono sempre reperibili dal Bosch Catalog: 
 
Percorso: Video Systems – Strumenti per la configurazione – ATE Script – Guida Script ATE 

 
 
 

 

 

 

 
 
 
 
 
 
 

https://www.plcqt.boschsecurity.com/BoschCatalog/0029/Marketing/Corporate%20Marketing/Bosch%20Catalog/Bosch-News-eCatalog_rev.1.pdf

